1974
USDA FOREST SERVICE GENERAL TECHNICAL REPORT PNW-27

AUTOMATED ANALYSIS OF TIMBER ACCESS ROAD ALTERNATIVES

Doyle Burke

Pacific Northwest Forest and Range Experimental Station
U.S. Department of Agriculture Forest Service Portland, Oregon

ABSTRACT

The evaluation of timber access road alternatives is one of the primary tasks in timber harvest planning and design. During the planning stages, it is also one of the most difficult to accomplish quantitatively because a basis for comparison is related to such values as grade, length, horizontal and vertical curvature, and volumes of excavation and embankment. Within reasonable time constraints, these values are almost impossible to produce with any accuracy. Evaluating several alternatives is usually out of the question.

The analysis of timber access road alternatives involves four analytic procedures:

1. Road projection of a proposedroad corridor with a specified grade across a topographic or orthophoto map,
2. horizontal alinement analysis of the projected road grade,
3. vertical alinement analysis of the horizontal alinement, and
4. computation of excavation and embankment quantities.

Automated analysis of these procedures has been implemented on a desk-top calculator system using digitizer and plotter peripherals. The logic and mathematics are discussed.

The analytic system presented is designed for preharvest planning and is based on map data. The analytic routines are structured to accept field data with minor programing changes.

Keywords: Road administration/planning (forest), computer programs, timber management planning.

INTRODUCTION

Timber harvest planning and design always begins with a thorough analysis of the maps and aerial photographs covering the proposed harvest area. This aspect of the design process can, if properly done, reveal much about the planning area and increase the efficiency of the layout crews in the field. Alternative roads, landings, and settings can be evaluated on paper, and the most suitable combination of these scheduled for field examination.

Much information about the timber stand and the terrain it occupies may be determined from maps and photographs. Quantitative values for area, volume, species, size, and distribution can be obtained by the experienced planner. Simple methods and procedures for doing this are readily available to the forest manager.

The quantitative evaluation of road and landing alternatives is much more complex. Bases for comparison are such values as grade, length, horizontal and vertical alinement, and volumes of excavation and embankment. These values are not readily apparent from the topographic maps but could be laboriously obtained manually if time allowed. Even so, examining several alternatives is usually out of the question.

This paper presents an analytic system for evaluating timber access road alternatives, using topographic maps and a desk-top calculator, digitizer, and plotter. The procedure is simple, fast, accurate, and allows maximum interaction and engineering judgment on the part of the designer. This information should be useful to resource managers and of particular interest in planning logging and transportation systems.

THE DESIGN PROCESS

The general design process for timber access roads can produce results useful in evaluating and comparing planning alternatives. The design flow of this process is illustrated in figure 1.

Each step of the design process produces results that are used to 1) determine if that alternative is satisfactory with respect to design standards, and 2) generate an array of information to be used in comparing alternate routes. The results obtained at the various steps of the design process are shown in figure 2.



Figure 1.—Design flow for timber access road analysis.



Figure 2.--Design results useful in evaluating timber access road alternatives.

THE DESIGN SYSTEM

A design system capable of high-speed data manipulation, maximum designer interaction, and ready availability is needed to facilitate rapid and accurate evaluation of timber access road alternatives. Desk-top calculator systems with digitizer and plotter facilities best meet these requirements,

Modern desk-top calculator systems (fig. 3) are capable of high-speed, complex numeric computations with data input and output through digitizers, plotters, and printers. Maximum designer interaction is possible because the designer brings his maps, photographs, and concept of the problem directly to the system and uses his skill, knowledge, and judgment to analyze the trial road, vary its parameters, and converge on an acceptable solution. The desk-top computer system relieves the designer of routine "number crunching" and allows him to apply professional level skills and judgment to the problem. This design system allows the designer to immediately see the results of his decisions, make desired changes, and carry out many more design cycles than are possible using manual methods or even some methods employing large-scale computer systems.

Figure 3.--Desk-top calculator system with digitizer, plotter, printer, and card reader.

The use of the digitizer in this design process is indispensable. The digitizer is necessary to convert the graphically displayed information on the topographic map to numeric coordinates (X, Y, Z). These coordinates are transferred to the computer and processed by the analytic routines. The results are then displayed on the plotter and printer for evaluation by the designer.

DESCRIPTION OF ANALYTIC ROUTINES

Road Projection

The road projection routine is used by the designer to project trial roads at selected grades from a topographic or orthophoto map base. The map is placed on the digitizer and trial locations projected using the digitizer cursor. Information from the digitizer is processed by the desk-top calculator, and an overlay plot of the routes is made on the plotter. The plots from this routine will subsequently be used by the horizontal alinement routine. The general logic for this routine is shown in figure 4.



Figure 4.—Macroflow chart for road projection routine.

Horizontal Alinement

This routine is used to develop the horizontal alinement data using the trial grade lines produced *bythe road projection routine. The plotted grade lines on the overlay are transferred to the map positioned on the digitizer. Using the map and the grade line overlay, P. I, points (point of intersection between tangents) are selected using the digitizer. This information is processed by the calculator, L-line data printed (curve data for each curve, stationing, number of curves, and total curvature) and plan view of the L-line plotted. The plots from this routine will be used by the vertical alinement routine. The general logic for the horizontal alinement routine is shown in figure 5.

The calculations for L-line stationing and curve data are similar to those used in standard highway engineering texts.[1/]

Vertical Alinement

The vertical alinement routine is used to introduce vertical curve geometry to the road grade and produce cut and fill depths' at the centerline of the road. A labeled plot of the road ground profile and grade profile is also produced.

The overlay generated by the horizontal alinement routine is used to determine ground elevations from the topographic base map on the digitizer. The mathematics for the vertical curve geometry used by the vertical alinement routine are similar to those used in standard highway engineering texts (e.g., see footnote 1).

The general logic for the vertical alinement routine is shown in figure 6.

Earthwork Routine

The earthwork routine combines cut and fill information from the vertical alinement routine with descriptions of the road template (road bed width, cut and fill slopes, and ground slopes) to produce excavation and embankment quantities. The macroflow chart for this routine is shown in figure 7.

The computations for roadway cross-section areas used in the earthwork routine are based on "slope stake" notation. This decision allows the routine to accept data from the digitizer, or in field note form with minor program changes, To accept data from the digitizer, slope stake points are located using slope intercept calculations.

1/ Thomas F. Hickerson. *Route surveys and design.* 4th edition, McGraw-Hill Book Co., Inc., New York, 1959.



Figure 5.—Macroflow chart for horizontal alinement routine.



Figure 6.—Macroflowchart for vertical alinement routine.



Figure 7.—Macroflow chart for earthwork analysis routine.

Excavation section (refer to fig. 8).--

$$\frac{H1 - C2}{S} = \frac{H1}{C} - R/2$$

$$H1 - C2 = \frac{H1(S)}{C} + R/2\ (S)$$

$$H1(C) - H1(S) = C2(C) + R/2\ (S)(C)$$

$$H1 = \frac{C2(C) + R/2(S)(C)}{C - S}$$

and, using the same geometry,

$$C2 - H2 = \frac{S(H2)}{C} + R/2\ (S)$$

$$C(C2) - C(H2) = S(H2) + R/2\ (C)(S)$$

$$H2 = \frac{C2(C) - R/2\ (C)(S)}{C + S}$$

and

$$X4 = \frac{H1 - C2}{S}$$

$$02 = X4 + R/2$$

$$02 = \frac{H1 - C2}{S} - R/2$$

$$X5 = \frac{H2}{C} + R/2$$

$$D1 = X5 + R/2$$

$$D1 = \frac{H2}{C} + R$$

Area of Excavation Section $= X = 0.5[(H1)(D1) + (H2)(D2)]$

Embankment section (refer to fig. 9).--

Using the same geometry,

$$H1 = [(C1)(F) + R/2\ (S)(F)] / (F-S)$$

$$H2 = [(C1)(F) - R/2\ (S)(F)] / (F+S)$$

$$D1 = (H2/F) + R$$

$$02 = [(H1 - C1) / S] + R/2$$

Area of Embankment Section = 0.5 $[H1(D1) + H2(D2)]$



Figure 8.--Finding area of excavation section.



Figure 9.--Finding area of embankment section.

Combination section.--In a combination section (one with excavation and embankment) there are two conditions to be treated: **(a)** fill at centerline, and (b) cut at centerline.

1. Fill at centerline (refer to fig. 10)

$$03 = \frac{C1}{S}$$

$$D1 = \frac{H1}{S}$$

$$D1 = \frac{H1}{C} + (R/2 - D3)$$

$$\frac{I^n}{S} = \frac{1}{C} + (R/2 - D3)$$

$$HI = \frac{H1(S)}{C} + S(R/2 - D3)$$

$$H1(C) = H1(S) + (R/2 - D3)(S)(C)$$

$$H1(C-S) = (R/2 - D3)(C)(S)$$

$$H1 = \frac{(R/2 - D3)(S)(C)}{(C - S)}$$

Area of excavation = [HI $(R/2 - D3)$] / 2

then,

$$D2(S) = D2(F) - (R/2 + D3)(F)$$

$$D2(S) - D2(F) = -(R/2 + D3)(F)$$

$$02 = \frac{(R/2 + D3)(F)}{F - S}$$

and

$$H2 = D2(S)$$

$$H2 = \frac{(R/2 + D3)(F)(S)}{F - S}$$

Area of embankment $= M = [(R/2 + D3)\ H2] / 2$



Figure 10.--Finding area of combination section with fill at centerline.

2. Cut at centerline (refer to fig. 11)

Using the same geometry as in fill,

$$03 = \frac{c2}{S}$$

$$D1 = \frac{H1}{C} + (R/2 + 03)$$

$$02 = H2/S$$

$$H1 = \frac{(R/2 + D3)(S)(C)}{C - S}$$

Area of excavation = $[H1\ (R/2 + D3)] / 2$

$$H2 = \frac{(R/2 - D3)(F)(S)}{F - S}$$

Area of embankment = $[(R/2 - 03)\ H2] / 2$



Figure 11.--Finding area of combination section with cut at centerline.

LIMITATIONS

The design system routines discussed in previous sections will treat the specific problem of evaluating timber access road alternatives. Although the program codes are designed for relative ease of modification and use, there are several limitations that should be discussed.

First, the analytic routines are specifically designed to produce quantitative information for evaluating planning alternatives. In this respect, topographic data are expected to come from topographic or orthophoto maps. The routines may, however, be modified to accept field data by altering the program input statements.

Second, the horizontal alinement routine uses curve radius rather than degrees. It is felt that' curvature is easier to visualize on maps using radii. The analytic routine provides for the degree function and could accept degree input.

Third, the roadway template does not provide for ditches, curve widening, superelevation, or curvature corrections for earthwork. For planning purposes, it is felt that these items make a relatively small contribution to volumes, while detailed road design will require their consideration. Adjustments for shrink or swell can be made from the earthwork summary.

The design of the analytic routines is general enough that the above limitations can be eliminated without major logic or programing changes. What changes are required can be accomplished by any competent BASIC language programer.

APPENDIX A

EXAMPLE PROBLEM OPERATING INSTRUCTIONS

Explanation

The purpose of this example is to illustrate the procedure for locating a road on a topographic map and then evaluating the proposed location in terms of horizontal alinement, vertical alinement, and earthwork. The operating instructions presume a knowledge of the Hewlett-Packard Model 9830A[2/] system configuration (fig. 12) and initializing procedures. Supplementary display instructions are provided under program control.

1. Install "Timber Access Road Pac" tape in the calculator. Place "Timber Access Road Pac" overlay (fig. 13) over S. F. keys.
2. Press LOAD, EXECUTE.
3. Press RUN, EXECUTE.
4. Select the desired program by pressing the appropriate S. F. key. Keep in mind that the earthwork program post-processes vertical alinement program data.
5. Refer to program instructions for program selected.
6. General instructions may be obtained by pressing S. F. key 4.

Road Projection

1. Press S.F. key 8.
2. Place the map on the digitizer (fig. 14).
3. Place transparent overlay material on the plotter. Adjust plotting area for 9 x 14 inches.
4. Enter map scale, <u>feet</u> per inch.
5. Enter contour interval, <u>feet</u>.
6. Digitize first tic mark (used to aline overlay on the map) after hearing beep, Press [S] on the cursor.
7. Digitize second tic mark. Press [S] on the cursor.
8. Enter trial grade, <u>percent</u>.
9. Digitize road starting point. Press [S] on the cursor.
10. Digitize the road location by moving the cursor to the next contour away from the starting point, press [C] on the cursor, and move the cursor along the contour until a beep is heard. Move the digitizer to the next contour and continue across the map along the desired road location. To change grade, press STOP, CONT 250, EXECUTE, and repeat instructions beginning at step 8. The plotter will plot the location on the overlay.
11. At the end of the road location, press STOP, SHIFT, and S. F. key 8.
12. The printer will print summary of input data and length of the road projection in feet and miles.
13. Follow display instructions for new trial road.

[2/] *The use of trade, firm, or corporation names in this publication is not an official endorsement or approval by the U.S. Department of Agriculture of any product to the exclusion of others which may be suitable.*

Figure 12.

MODEL 9830A SYSTEM CONFIGURATION

MODEL 30 CALCULATOR

☐ 1760 Words RWM, Basic Calculator
☒ 3808 Words RWM, Option 275
☐ 7904 Words RWM, Option 276

ROMS

Plotter Control 271	
String Variables 274	
Extended 1/0 272	

PERIPHERALS	SELECT CODE
0 9 8 A Card Reader..	
☐ 9861A Typewriter.	
☒ 9862A Plotter.	Primary
☐ 9863A Paper Tape Reader.	
☒ 9864A Digitizer..	Primary
☒ 9865A Cassette Memory.	5
☒ 9866A Thermal Page Printer.	Primary
☐ 9867 Mass Memory Drive (Unit No.)	
☐	
☐	
☐	
☐	
☐	



Figure 13.--Special function key overlay ***card.***



Figure 14.--Map ***positioned on the*** *digitizer.*

Horizontal Alinement

1. Press S. F. key 1.
2. Place road projection plot on the map (fig. 15).
3. Place transparent overlay material on the plotter, Use 9- x 14-inch plotting area.
4. Enter map scale, feet per inch.
5. Plotter will label plot.
6. Digitize starting point of the road. Use topographic map and overlay plot to select point. Press [S] on the cursor.
7. Digitize the next P.I. point. Use the map and overlay plot to select the appropriate point. Press [S] on the cursor. The plotter will plot the tangent.
8. Digitize the next P.I. point.
9. Enter curve radius, feet.
10. Enter direction, right or left.

11. The plotter will identify curve and plot curve. Printer will list the curve data:

- Delta
- Radius
- Tangent
- Length
- P.C. station
- P.T. station
- P.I. equation

12. Repeat at step 8, for remaining road length.

13. At the end of the road, press STOP, SHIFT, S. F. key 1. The printer will list a summary giving total number of curves, total curvature, and L-line distance in feet and miles.



*Figure 15.--Road projection plot **positioned on** the map.*

Vertical Alinement

1. Place horizontal alinement plot on the topographic map (fig. 16).
2. Place cross-section paper on the plotter. Use 9- x 14-inch plotting area and install tape cassette in cassette memory device premarked with appropriate number of 20-word files.
3. Enter map scale, feet per inch.
4. Enter contour interval, feet.
5. Enter starting elevation of road, feet.

6. Digitize road profile. Digitize each point where a contour crosses the L-line on the overlay, using [S] on the cursor. Begin at station 0+00 and use the S. F. keys 5 and 6 to change ascending to descending points, and vice versa. The program assumes ascending points to start.

7. When profile is digitized, press STOP, S. F. key 7.

8. Enter desired horizontal and vertical scale of the plot; feet, feet.

9. The plotter will label the plot and plot output. The printer will list distance and elevation.

10. For analysis of cut or fill at the centerline, press S. F. key 8. This provides a routine to allow the introduction of parabolic vertical curves.

11. Enter grade elevation at station 0; feet, 0, EXECUTE.

12. Enter first P. V. I. (point of vertical intersection) elevation and station; ~~feet~~, f~~eet~~, EXECUTE.

13. Enter next P. V. I. elevation and station; ~~feet~~, ~~feet~~, EXECUTE.

14. Enter vertical curve length in stations; e. g., 600 feet = 6 stations. Plotter will plot road grade and the printer will list cut and fill data. The data will be stored on tape.

15. Repeat at step 13 until the end of the road is encountered.

16. Press SHIFT, S. F. key 2.

17. The plotter will complete the plot (fig. 17), and the printer will complete the listing and print a summary of vertical curve data.



Figure 16.--Horizontal alinement plot positioned on map with partial printer listing.



VERTICAL CURVE DATA

STATION		GROUND	GRADE	CUT	FILL
0		2600.0	2600.0	0	0
0.0	+ 10.8	2610.0	2601.4	8.6	
0.0	+ 44.8	2620.0	2606.0	14.0	
0.0	+ 82.3	2630.0	2611.0	19.0	
1.0	+ 22.3	2640.0	2616.3	23.7	
1.0	+ 48.0	2650.0	2619.7	30.3	
1.0	+ 87.7	2660.0	2625.0	35.0	
3.0	+ 75.7	2670.0	2650.1	19.9	
5.0	+ 7.3	2680.0	2667.6	12.4	
6.0	+ 27.3	2690.0	2683.6	6.4	
7.0	+ 48.9	2700.0	2699.4	0.6	
7.0	+ 73.3	2710.0	2702.4	7.6	
8.0	+ 11.7	2720.0	2707.2	12.8	
9.0	+ 74.9	2730.0	2727.0	3.0	
10.0	+ 4.6	2740.0	2730.6	9.4	
10.0	+ 17.4	2750.0	2732.1	17.9	
11.0	+ 43.7	2760.0	2747.2	12.8	
12.0	+ 47.4	2770.0	2759.7	10.3	
14.0	+ 62.3	2780.0	2785.5		5.5
15.0	+ 64.3	2790.0	2797.7		7.7
15.0	+ 97.1	2800.0	2801.7		1.7
16.0	+ 18.4	2810.0	2804.2	5.8	
16.0	+ 48.1	2820.0	2807.8	12.2	
17.0	+ 0.6	2830.0	2814.2	15.8	
17.0	+ 66.6	2840.0	2822.1	17.9	
18.0	+ 11.1	2850.0	2827.5	22.5	
19.0	+ 29.1	2860.0	2841.7	18.3	
20.0	+ 70.5	2870.0	2858.8	11.2	
21.0	+ 55.8	2880.0	2869.1	10.9	
22.0	+ 95.8	2890.0	2886.7	3.3	
24.0	+ 19.5	2900.0	2903.6		3.6
25.0	+ 59.2	2910.0	2924.3		14.3
26.0	+ 7.6	2920.0	2931.8		11.8
26.0	+ 46.4	2930.0	2938.0		8.0
26.0	+ 78.0	2940.0	2943.2		3.2
27.0	+ 15.2	2950.0	2949.3	0.7	
27.0	+ 50.8	2960.0	2955.1	4.9	
28.0	+ 38.7	2970.0	2969.7	0.3	
29.0	+ 13.7	2980.0	2982.4		2.4
29.0	+ 90.0	2990.0	2995.7		5.7
30.0	+ 43.8	3000.0	3005.3		5.3
31.0	+ 30.1	3010.0	3021.0		11.0
31.0	+ 62.6	3020.0	3027.0		7.0
32.0	+ 8.0	3030.0	3035.5		5.5
32.0	+ 47.7	3040.0	3043.0		3.0
32.0	+ 84.5	3050.0	3050.0		0.0

STATION	P.V.I. ELEV.	V.C. LENGTH
750.0	2700.0	3.0
1600.0	2802.0	4.0
2420.0	2901.0	5.0
3050.0	3004.0	6.0

Figure 17.--Completed road profile plot with partial printer listing.

Earthwork

1. Press **S.F.** key 3.
2. Rewind data tape on cassette memory.
3. Enter road bed width, <u>feet</u>, EXECUTE.
4. Enter cut slope, <u>value</u>, EXECUTE: 1 (<u>1</u> = 1:1).
5. Enter fill slope, <u>value</u>, EXECUTE: 1 (<u>1.5</u> = 1.5:1).
6. Enter number of side slope breaks to be used (maximum of 20), <u>value</u>, EXECUTE.
7. Enter each side slope and ending station, <u>side slope</u>, <u>ending station</u>, EXECUTE. Enter side slope as a decimal (e. g., 25 percent = 0.25).
8. The printer will list results of earthwork analysis and summary of excavation and embankment (fig. 1.8).

```
ROADBED WIDTH = 16.0     FEET
CUT SLOPE RATIO = 1.0  :1
FILL SLOPE RATIO = 1.5  :1
SIDE SLOPE OF 45.0    PERCENT ENDS AT STA. 1500.0
SIDE SLOPE OF 30.0    PERCENT ENDS AT STA. 2200.0
SIDE SLOPE OF 50.0    PERCENT ENDS AT STA. 4300.0
```

EARTHWORK QUANTITIES

	CENTER-LINE		AREA (SQ.FT.)		VOLUME (CU.YD)	
STATION	CUT	FILL	EXC	EMB	EXC	EMB
0.0	0.0	0.0	26.2	26.2	0.0	0.0
312.0	0.0	11.2	0.0	295.9	151.3	1860.9
446.2	0.0	4.6	0.0	110.0	0.0	1008.5
591.6	0.8	0.0	39.8	24.8	107.3	363.0
780.4	2.0	0.0	62.5	18.4	357.6	151.0
985.9	2.2	0.0	68.7	16.1	498.9	131.5
1216.5	2.5	0.0	73.9	14.1	609.0	128.9
1432.4	4.9	0.0	197.3	0.0	1084.4	56.2
1699.5	3.3	0.0	138.6	0.0	1661.4	0.0
1977.0	0.8	0.0	23.9	12.3	835.1	63.2
2233.0	0.1	0.0	33.8	32.0	273.6	209.9
2450.8	2.8	0.0	92.8	16.2	510.4	194.3
2711.0	2.3	0.0	78.6	21.7	825.8	182.6
2975.9	1.4	0.0	57.7	28.2	668.6	245.0
3234.2	1.0	0.0	49.6	30.1	513.1	279.0
3501.8	0.0	0.0	32.3	32.0	405.9	307.6
3860.7	0.0	7.6	0.0	192.6	214.8	1492.5
4108.7	0.0	6.6	0.0	164.8	0.0	1641.8
4283.6	0.0	0.0	32.6	32.0	105.7	637.3

```
TOTAL EMBANKMENT = 8953.2  CUBIC YARDS.
TOTAL EXCAVATION = 8822.9  CUBIC YARDS.
```

Figure 18.--Printer listing of earthwork analysis results.

APPENDIX B

PROGRAM LISTINGS

VARIABLES USED IN THE ROAD PROJECTION PROGRAM, TAPE FILE 2

H = Horizontal Map Scale
C = Contour Interval
Y = Tic Mark Coordinates
G = Trial Grade, Percent
M = Computed Map Distance
X1
Y1 = Initial Coordinates
X2
Y2 = Intermediate Coordinates
R = Trial Distance
R1 = Accumulate Road Distance

VARIABLES USED IN THE HORIZONTAL ALINEMENT PROGRAM, TAPE FILE 3

A$ = Alphameric String for Curve Direction
N = Curve Number
I = Plotting Counter
Z = Degree Accumulator
H = Horizontal Map Scale
X
Y = Operating Coordinates
X1
Y1 = Coordinates of Road Starting Point
X2
Y2 = Back Tangent Coordinates
X3
Y3 = Forward Tangent Coordinates
01 = Angle of Back Tangent
S1 = Back Stationing at P.I.
s2 = Length of Back Tangent
02 = Angle of Forward Tangent
A = Polar Angle
D2 = Length of Forward Tangent
0 = Angle Between Tangents
D = Variable for Quadrant Correction
R1 = Curve Radius
T = Curve Tangent
R2 = Degree of
L = Curve Length
P1 = P. C. Distance
P2 = P. T. Distance
P8
P9 = Normalized Values of P. C. Station
P7
P6 = Normalized Values of P. T. Station

S8 S9 S7 S6	=	Normalized Values of P. I. Equation
Z2	=	Distance Accumulator
R	=	Polar Distance
D5	=	Central Angle Plot Increment
D6	=	Deflection Angle
D7	=	Deflection Angle Increment

VARIABLES USED IN THE VERTICAL ALINEMENT PROGRAM, TAPE FILE 4

AS(I, 1)	=	Station
AS(I, 2)	=	Ground Elevation
B(1)	=	Station
B(2)	=	Ground Elevation
B(3)	=	Grade Elevation
CØ	=	Counter
I	=	Array Index Variable
S1	=	Distance Accumulator
J	=	File Counter
K	=	Array Index Variable
VS(20, 3)	=	P. V. I. Station and Elevation Array
H	=	Horizontal Map Scale
C	=	Contour Interval
E5	=	Starting Elevation
E	=	Elevation Print Variable and Accumulator
X Y	=	Operating Coordinates
X1 Y1	=	Initial Coordinates
X2 Y2	=	Forward Coordinates
S	=	Distance Between Coordinates
V	=	Vertical Plot Scale
X	=	Lable Variable
A	=	Array Index Variable
E2	=	Starting Grade Elevation
S2	=	Starting Station
G1	=	Back Tangent Grade
E3	=	Back P. V. I. Elevation
S3	=	Back P. V. I. Station
E4	=	Forward P. V. I. Elevation
S4	=	Forward P. V. I. Station
G2	=	Forward Tangent Grade
L1	=	Vertical Curve Length
P1	=	P. V. C.
P2	=	P. V. T.

S1 S5	=	Normalized Stationing
	=	Grade Elevation
R2	=	Rate of Change of Grade
C1	=	Cut or Fill at Centerline

VARIABLES USED IN THE EARTHWORK .PROGRAM, TAPE FILE 5

AS(100, 2)	=	Cornmon Buffer
BS(1)	=	Station
BS(2)	=	Ground Elevation
BS(3)	=	Grade Elevation
D(20, 2)	=	Side Slope Break Array
CØ	=	Counter
Q	=	Number of Side Slope Breaks
X3	=	Excavation Accumulator
M3	=	Embankment Accumulator
X2	=	Excavation Volume Between Sections
M2	=	Embankment Volume Between Sections
X	=	Area of Excavation
M	=	Area of Embankment
A	=	File Load Counter
R	=	Roadbed Width
C	=	Cut Slope
F	=	Fill Slope
S	=	Slide Slope
J	=	File Counter
L	=	Centerline Stationing
C1	=	Centerline Fill (Combination Section)
C2	=	Centerline Cut (Combination Section)
D3	=	Offset Distance (Template)
D1 D2	=	Template Distances
H1 H2	=	Template Heights

TAPE FILE 0

```
10 REM LOAD SPECIAL FUNCTION KEYS, TAPE FILE 0
23 LOAD  KEY 1
30 DISP "SELECT PROGRAM ON SF KEYS";
40 END
```

TAPE FILE 1, SPECIAL FUNCTION KEYS

SF KEY 0

```
*LOAD2,10,10*
```

SF KEY 1

```
*LOAD3,10,10*
```

SF KEY 2

```
*LOAD4,1,1*
```

SF KEY 3

LOAD5,1,1

SF KEY 4

```
10 PRINT "GENERAL OPERATING INSTRUCTIONS"
20 PRINT "----------------------------------------------------------------------"
30 PRINT "THIS TIMBER ACCESS ROAD PACKAGE REQUIRES THE DIGITIZER, PLOTTER, AND"
40 PRINT "OUTBOARD TAPE CASSETTE.  THE PLOTTER SHOULD BE SET FOR A PLOTTING"
50 PRINT "AREA OF 9 IN. X 14 IN. AND YOU SHOULD HAVE AVAILABLE BOTH  TRANSPARENT
60 PRINT "AND CROSS-SECTION PAPER.  THE ROAD PROJECTION AND HORIZONTAL "
70 PRINT "ALIGNMENT ROUTINES REQUIRE THE FORMER, AND THE VERTICAL ALIGNMENT"
80 PRINT "ROUTINE REQUIRES THE LATTER."
90 PRINT
100 PRINT "YOU SHOULD ALSO HAVE AVAILABLE AN EARTHWORK DATA TAPE PREMARKED"
110 PRINT "WITH AN APPROPRIATE NUMBER OF 20 WORD FILES (SAY 200)."
120 PRINT
130 PRINT "SPECIFIC OPERATING INSTRUCTIONS ARE PROVIDED IN THE WRITTEN "
140 PRINT "DOCUMENTATION.  THE SPECIAL FUNCTION KEYS, ALONG WITH VISUAL"
150 PRINT "PROMPTING FROM THE DISPLAY AND PRINTER OUTPUT WILL PROVIDE      "
160 PRINT "SUFFICIENT INSTRUCTIONS ONCE THE OPERATIOR IS FAMILIAR WITH THE"
170 PRINT "PACKAGE."
180 PRINT
190 PRINT "REMEMBER, THE EARTHWORK ROUTINE PROCESSES DATA COMPILED BY THE"
200 PRINT "VERTICAL ALIGNMENT ROUTINE!   IT WILL PROCESS PREVIOUSLY COMPILED"
210 PRINT "EARTHWORK DATA TAPES."
220 PRINT
230 PRINT "NOW, PROCEED WITH THE SELECTION OF THE APPROPRIATE PROGRAM."
240 PRINT
250 PRINT
260 DISP "SELECT PROGRAM ON SF KEYS"
270 END
```

SF KEY 5

CONT275

SF KEY 6

CONT460

SF KEY 7

CONT2001

SF KEY 8

CONT3001

SHIFTED SF KEY 0

CONT470

SHIFTED SF KEY 1

CONT 690

SHIFTED SF KEY 2

CONT3280

TAPE FILE 2

```
10 REM ROAD PROJECTION ROUTINE,TAPE FILE 2
20 REM PROGRAMMED BY BURKE,PNW,8/73
21 PRINT "ROAD PROJECTION ROUTINE"
24 PRINT "WHEN FINISHED, PRESS-STOP-SHIFT-SF KEY 0."
25 PRINT
26 FIXED 1
30 DISP "MAP SCALE =";
40 INPUT H
50 PRINT "MAP SCALE="H"FEET/INCH"
60 DISP "CONTOUR INTERVAL=";
70 INPUT C
80 PRINT "CONTOUR INTERVAL="C"FEET"
90 SCALE 0,14*H,0,9*H
100 DISP "DIGITIZE 1ST TIC MARK"
105 WAIT 2000
110 WRITE (9,*)
120 WAIT 1000
130 ENTER (9,*)X,Y
140 GOSUB 1000
150 DISP "DIGITIZE 2ND TIC MARK"
155 WAIT 2000
160 WRITE (9,*)
170 WAIT 1000
180 ENTER (9,*)X,Y
190 GOSUB 1000
200 R1=0
210 PRINT
220 PRINT "******************************"
230 PRINT
240 DISP "ENTER TRIAL GRADE (%)";
250 INPUT G
260 PRINT "TRIAL GRADE ="G"%"
270 M=C/(H*G*0.01)
280 DISP "DIGITIZE STARTING POINT"
290 WAIT 1000
300 ENTER (9,*)X1,Y1
310 GOSUB 2000
320 DISP "DIGITIZE TRIAL GRADE LINE"
330 WAIT 1000
335 DISP "PRESS-STOP-CONT470-WHEN DONE"
336 WAIT 3000
340 ENTER (9,*)X2,Y2
350 X=X2-X1
360 Y=Y2-Y1
370 R=SQR(X*X+Y*Y)
380 IF R<M THEN 340
390 WRITE (9,*)
400 WAIT 100
410 WRITE (9,*)
```

```
420 GOSUB 3000
430 R1=R1+R
440 X1=X2
450 Y1=Y2
460 GOTO 340
470 PRINT "GRADE LINE DISTANCE="R1*H"FEET, OR"R1*H/5280"MILES.
480 PRINT
485 DISP "PRESS-CONT200-FOR NEW TRIAL ROAD"
490 STOP
1000 PEN
1010 PLOT X*H,Y*H,-2
1020 CPLOT -0.3,-0.3
1030 LABEL (*)"+"
1040 PEN
1050 RETURN
2000 X=X1
2010 Y=Y1
2020 PLOT X*H,Y*H,-2
2030 RETURN
3000 PLOT X2*H,Y2*H,-2
3010 RETURN
3020 END
```

TAPE FILE 3

```
10 REM  HORIZONTAL ALIGNMENT PROGRAM (HAL), TAPE FILE
20 REM  PROGRAMMED BY BURKE, PNW, 8/73
21 DIM A$[10]
22 N=I=Z=0
25 DEG
30 FIXED 0
31 PRINT " HORIZONTAL ALIGNMENT ROUTINE"
40 DISP "MAP SCALE=";
50 INPUT H
60 PRINT "MAP SCALE="H" FEET/INCH"
61 PRINT "PRESS -STOP-SHIFT-SF KEY 1- WHEN FINISHED"
62 PRINT
63 PRINT
70 SCALE 0,14*H,0,9*H
71 PLOT H,9*H,1
72 LABEL (*)"HORIZONTAL ALIGNMENT PLOT"
73 LABEL (*)"SCALE ="H"FEET PER INCH"
80 DISP "DIGITIZE STARTING POINT"
90 WAIT 2000
95 WRITE (9,*)
100 ENTER (9,*)X1,Y1
110 X=X1
120 Y=Y1
125 PEN
130 PLOT X*H,Y*H,-2
140 DISP "DIGITIZE NEXT P.I. POINT
150 WAIT 1000
```

```
160 WRITE (9,*)
170 ENTER (9,*)X2,Y2
180 X=X2
190 Y=Y2
200 GOSUB 1010
210 X=X2-X1
220 Y=Y2-Y1
230 GOSUB 2000
240 O1=A
250 S1=0
260 S2=R*H
270 DISP "DIGITIZE NEXT P.I. POINT"
280 WAIT 1000
290 WRITE (9,*)
300 ENTER (9,*)X3,Y3
310 X=X3
320 Y=Y3
330 GOSUB 1010
340 X=X3-X2
350 Y=Y3-Y2
360 GOSUB 2000
370 O2=A
380 D2=R*H
390 O=ABS(O2-O1)
400 IF O<180 THEN 420
410 D=360-O
415 GOTO 430
420 D=O
430 DISP "CURVE RADIUS & DIRECTION=";
440 INPUT R1,A$
450 T=R1*TAN(D/2)
460 R2=5730/R1
470 L=(D/R2)*100
480 P1=S2-T
490 P2=P1+L
500 S1=P2-T
505 N=N+1
506 GOSUB 3500
510 PRINT
511 Z=Z+D
520 PRINT "CURVE DATA FOR CURVE NO."N
530 WRITE (15,5000)D
531 IF A$="RIGHT" THEN 540
535 WRITE (15,5002)R1
536 GOTO 550
540 WRITE (15,5001)R1
550 WRITE (15,5003)T
560 WRITE (15,5004)L
561 P8=INT(P1/100)
562 P9=P1-(P8*100)
563 P7=INT(P2/100)
564 P6=P2-(P7*100)
565 S8=INT(S1/100)
566 S9=S1-(S8*100)
567 S7=INT(S2/100)
```

```
568 S6=S2-(S7*100)
570 WRITE (15,5005)P8,P9
580 WRITE (15,5006)P7,P6
590 WRITE (15,5007)S8,S9,S7,S6
610 PRINT
620 PRINT
630 GOSUB 3000
640 Y2=Y3
650 X2=X3
660 O1=O2
670 S2=S1+D2
680 GOTO 270
690 PRINT "+++++++++++++++++++++++++++++++++"
691 PRINT "HORIZONTAL ALIGNMENT SUMMARY"
692 PRINT "TOTAL NUMBER OF CURVES ="N
693 PRINT "TOTAL CURVATURE ="Z"DEGREES"
694 Z2=S1+(D2-T)
695 PRINT "TOTAL L-LINE DISTANCE ="Z2"FEET"
696 FIXED 2
697 PRINT "OR"Z2/5280"MILES."
698 STOP
1000 PEN
1010 PLOT X*H,Y*H,-2
1020 RETURN
2000 R=SQR(X*X+Y*Y)
2010 A=ATN(Y/(X+(1E-90)*(X=0)))+2*SGNY*ATN1E+99*(X<0)
2020 RETURN
3000 PEN
3010 X=X2+((T/H)*COS(O1+180))
3020 Y=Y2+((T/H)*SIN(O1+180))
3030 GOSUB 1010
3040 IF A$="LEFT" THEN 3070
3050 D5=-D/16
3060 GOTO 3080
3070 D5=D/16
3080 D6=O1+D5
3090 X=X+(((L/8)/H)*COSD6)
3100 Y=Y+(((L/8)/H)*SIND6)
3110 GOSUB 4010
3120 I=I+1
3130 IF I=8 THEN 3170
3140 D7=2*D5
3150 D6=D6+D7
3160 GOTO 3090
3170 X=X3
3180 Y=Y3
3190 GOSUB 1000
3191 I=0
3200 RETURN
3500 PLOT X2*H,Y2*H,1
3510 CPLOT -1,-1
3520 LABEL (*)N
3530 PEN
3540 RETURN
4000 PEN
```

```
4010 PLOT X*H,Y*H,+2
4020 RETURN
5000 FORMAT "DELTA =",F4.0," DEGREES"
5001 FORMAT "RADIUS =",F4.0," FEET RIGHT"
5002 FORMAT "RADIUS =",F4.0," FEET LEFT"
5003 FORMAT "TANGENT =",F4.0," FEET"
5004 FORMAT "LENGTH =",F4.0," FEET"
5005 FORMAT "P.C.= STA.",F4.0,"+",F3.0
5006 FORMAT "P.T.= STA.",F4.0,"+",F3.0
5007 FORMAT "P.I. EQ.: STA.",F4.0,"+",F3.0," AHD.= STA.",F4.0,"+",F3.0,".BK."
6000 END
```

TAPE FILE 4

```
1 COM A$[100,2],B$[3],C0
10 REM VERTICAL ALIGNMENT ROUTINE, TAPE FILE 4
20 REM  PROGRAMMED BY BURKE,PNW, 8/73
30 FIXED 0
40 I=S1=J=K=0
41 DIM V$[20,3]
50 PRINT "VERTICAL ALIGNMENT ROUTINE"
51 PRINT "USE SF KEYS 5-8 FOR VARIOUS OPERATIONS"
52 PRINT "*******INSTALL EARTHWORK DATA TAPE ON OUTBOARD CASSETTE*********"
60 DISP "MAP SCALE =";
70 INPUT H
80 PRINT "MAP SCALE ="H"FEET PER INCH"
90 DISP "CONTOUR INTERVAL=";
100 INPUT C
110 PRINT "CONTOUR INTERVAL ="C"FEET"
120 DISP "STARTING ELEVATION =";
130 INPUT E5
131 E=E5
132 GOSUB 1000
140 PRINT "STARTING ELEVATION ="E"FEET"
150 PRINT
155 PRINT "-----------------------------"
160 DISP "DIGITIZE PROFILE POINTS"
170 WAIT 1000
180 ENTER (9,*)X1,Y1
185 REM     ASSUMES UP
190 ENTER (9,*)X2,Y2
200 X=X2-X1
210 Y=Y2-Y1
220 S=SQR(X*X+Y*Y)*H
225 S1=S1+S
230 E=E+C
240 GOSUB 1000
250 X1=X2
260 Y1=Y2
270 GOTO 190
275 REM       UP TO DOWN
280 DISP "ELEVATION =";
290 INPUT E
```

```
300 ENTER (9,*)X2,Y2
310 X=X2-X1
320 Y=Y2-Y1
330 S=SQR(X*X+Y*Y)*H
340 S1=S1+S
350 GOSUB 1000
360 X1=X2
370 Y1=Y2
380 ENTER (9,*)X2,Y2
390 X=X2-X1
400 Y=Y2-Y1
410 S=SQR(X*X+Y*Y)*H
420 E=E-C
430 S1=S1+S
440 GOSUB 1000
450 GOTO 360
460 REM       DOWN TO UP
470 DISP "ELEVATION =";
480 INPUT E
490 ENTER (9,*)X2,Y2
500 X=X2-X1
510 Y=Y2-Y1
520 S=SQR(X*X+Y*Y)*H
530 S1=S1+S
540 GOSUB 1000
550 X1=X2
560 Y1=Y2
570 GOTO 190
1000 I=I+1
1010 A[I,1]=S1
1020 A[I,2]=E
1040 RETURN
2001 DEG
2010 DISP "ENTER HOR.& VERT. SCALE";
2020 INPUT H,V
2030 SCALE -H/9,13.5*H,E5-V/9,E5+9*V
2031 PLOT H,E5+8.5*V
2032 LABEL (*)"ROAD PROFILE PLOT"
2033 LABEL (*,1.3,1.7,0,9/14)"HOR. SCALE: 1 IN.="H"FEET"
2034 LABEL (*,1.3,1.7,0,9/14)"VERT. SCALE: 1 IN.="V"FEET"
2040 OFFSET H/2,V/2
2045 XAXIS E5,H,0,13*H
2050 YAXIS 0,V,E5,E5+8*V
2055 OFFSET H/2,-V/2
2065 FOR X=0 TO 13*H STEP H
2070 PLOT X,E5+40,-1
2075 CPLOT 0,-0.3
2080 LABEL (*,1,1.7,90,9/14)X
2085 NEXT X
2090 OFFSET -H/2,V/2
2095 FOR Y=E5 TO E5+8*V STEP V
2100 PLOT 200,Y,-1
2105 CPLOT 0,-0.3
2110 LABEL (*,1,1.7,0,9/14)Y
2115 NEXT Y
2140 OFFSET H/2,V/2
```

```
2150 FOR A=1 TO I
2160 PLOT A[A,1],A[A,2]
2170 PRINT A[A,1];A[A,2]
2180 NEXT A
2185 PEN
2186 C0=I
2190 STOP
3000 REM    VERTICAL CURVE ROUTINE
3001 I=1
3005 FIXED 1
3010 PRINT "                         VERTICAL CURVE DATA"
3011 PRINT
3013 PRINT TAB5"STATION"TAB21"GROUND"TAB31"GRADE"TAB41"CUT"TAB51"FILL"
3014 PRINT
3020 DISP "ENTER STARTING ELEV.(STA.0+00)";
3030 INPUT E2,S2
3031 PRINT "    0     "TAB20,E2;TAB30,E2;TAB40"0"TAB50"0"
3032 PLOT S2,E2,-2
3036 B[1]=S2
3037 B[2]=E2
3038 B[3]=E2
3039 GOSUB 4030
3040 DISP "ENTER PVI ELEV.& STA.";
3050 INPUT E3,S3
3051 K=K+1
3052 V[K,1]=S3
3053 V[K,2]=E3
3060 G1=((E3-E2)/(S3-S2))*100
3070 DISP "ENTER NEXT PVI ELEV.& STA.";
3080 INPUT E4,S4
3090 G2=((E4-E3)/(S4-S3))*100
3100 DISP "ENTER V.C. LENGTH (STATIONS)";
3109 INPUT L1
3110 V[K,3]=L1
3111 K=K+1
3112 V[K,1]=S4
3113 V[K,2]=E4
3120 P1=(S3-(L1/2)*100)
3130 P2=(S3+(L1/2)*100)
3140 I=I+1
3141 IF A[I,1]>P2 THEN 3240
3150 S1=INT(A[I,1]/100)
3151 S5=A[I,1]-(S1*100)
3160 E=A[I,2]
3170 IF A[I,1]>P1 THEN 3220
3180 R1=E2+(A[I,1]-S2)*G1*0.01
3185 PLOT A[I,1],R1,-2
3190 IF R1<E THEN 3215
3200 C1=R1-E
3201 PRINT S1;TAB10,"+"S5;TAB20,E;TAB30,R1;TAB50,C1
3205 GOSUB 4000
3210 GOTO 3140
3215 C1=E-R1
3216 PRINT S1;TAB10,"+"S5;TAB20,E;TAB30,R1;TAB40,C1
3217 GOSUB 4000
```

```
3218 GOTO 3140
3220 X=(A[I,1]-P1)*0.01
3221 R2=(G2-G1)/L1
3222 R1=E2+((P1-S2)*G1*0.01)+(G1+0.5*R2*X)*X
3230 GOTO 3185
3240 E2=E3
3241 E3=E4
3249 S2=S3
3250 S3=S4
3260 G1=G2
3265 I=I-1
3270 GOTO 3070
3280 I=I+1
3285 E=A[I,2]
3290 R1=E2+(A[I,1]-S2)*G1*0.01
3295 PLOT A[I,1],R1,-2
3300 S1=INT(A[I,1]/100)
3310 S5=A[I,1]-(S1*100)
3320 IF R1<E THEN 3360
3330 C1=R1-E
3340 PRINT S1;TAB10,"+"S5;TAB20,E;TAB30,R1;TAB50,C1
3345 GOSUB 4000
3350 GOTO 3380
3360 C1=E-R1
3370 PRINT S1;TAB10,"+"S5;TAB20,E;TAB30,R1;TAB40,C1
3375 GOSUB 4000
3380 I=I+1
3390 GOTO 3285
4000 B[1]=A[I,1]
4010 B[2]=A[I,2]
4020 B[3]=R1
4030 J=J+1
4040 STORE  DATA #5,J,B
4045 IF J=C0 THEN 5010
4050 RETURN
5010 PEN
5011 PRINT
5012 PRINT
5013 PRINT "STATION",TAB15,"P.V.I. ELEV.",TAB30,"V.C. LENGTH"
5014 PRINT "========================================="
5015 FOR N=1 TO K-1
5016 PRINT V[N,1],TAB16,V[N,2],TAB33,V[N,3]
5017 NEXT N
5018 STOP
5020 END
6000 J=0
6010 J=J+1
6020 LOAD  DATA #5,J,B
6030 PRINT B[1];B[2];B[3]
6040 GOTO 6010
6050 STOP
```

TAPE FILE 5

```
1 COM A$[100,2],B$[3],C0
2 REM      EARTHWORK ANALYSIS PROGRAM, TAPE FILE 5
3 REM      PROGRAMMED BY BURKE, PNW, 12/73,3/74
10 FIXED 1
20 X3=M3=C1=C2=X2=M2=X=M=A=K=0
21 J=1
30 DIM D$[20,2]
40 DISP "ENTER ROADBED WIDTH";
50 INPUT R
60 DISP "ENTER CUT SLOPE";
70 INPUT C
80 DISP "ENTER FILL SLOPE";
90 INPUT F
100 DISP "HOW MANY SIDE SLOPE BREAKS";
102 INPUT Q
104 FOR I=1 TO Q
106 DISP "ENTER SIDE SLOPE AND ENDING STA.";
108 INPUT D[I,1],D[I,2]
110 NEXT I
115 PRINT
120 PRINT "ROADBED WIDTH ="R;"FEET"
130 PRINT "CUT SLOPE RATIO ="C":1"
140 PRINT "FILL SLOPE RATIO ="F":1"
150 FOR I=1 TO Q
152 PRINT "SIDE SLOPE OF"D[I,1]*100;"PERCENT ENDS AT STA."D[I,2]
153 NEXT I
160 PRINT
161 PRINT TAB30"EARTHWORK QUANTITIES"
162 PRINT
165 PRINT TAB13"CENTER-LINE"TAB32"AREA (SQ.FT.)"TAB52"VOLUME (CU.YD)"
170 PRINT "STATION"TAB11"CUT"TAB21"FILL"TAB32"EXC"TAB42"EMB"TAB52"EXC"TAB62"EMB"
180 PRINT
185 K=K+1
190 LOAD  DATA #5,J,B
195 A=A+1
196 IF A>C0 THEN 1002
199 IF B[1]>D[K,2] THEN 205
200 D1=B[1]
201 S=D[K,1]
203 GOTO 210
205 K=K+1
207 S=D[K,1]
210 IF (B[2]+(S*R/2))<B[3] AND (B[2]-(S*R/2))<B[3] THEN 440
220 IF B[2]+(S*R/2)>B[3] AND B[2]-(S*R/2)>B[3] THEN 600
230 REM SIDEHILL SECTION
240 IF B[2]>B[3] THEN 280
250 C1=B[3]-B[2]
260 D3=C1/S
270 GOTO 300
280 C2=B[2]-B[3]
290 D3=C2/S
300 X=((R/2+D3)*(-C*(R/2+D3))/(1-C/S))/2
```

```
310 M=((R/2+D3)*(-C*(R/2-D3))/(1-C/S))/2
320 IF J=1 THEN 370
330 X2=(((X+X1)/2)*(B[1]-L))/27
340 M2=(((M+M1)/2)*(B[1]-L))/27
350 X3=X3+X2
360 M3=M3+M2
370 GOSUB 1000
380 C1=C2=X2=M2=0
390 L=B[1]
400 X1=X
410 M1=M
420 J=J+1
425 IF J>C0 THEN 1002
430 GOTO 190
440 REM        FILL SECTION
450 C1=B[3]-B[2]
460 D1=C1+((R/2)*F)/(F+S)
470 D2=C1+((R/2)*F)/(F-S)
480 H1=S*D1
490 H2=S*D2
500 M=0.5*(C1*(D1+D2)+(R/2*(H1+H2)))
505 X=0
510 IF J=1 THEN 540
520 M2=(((M+M1)/2)*(B[1]-L))/27
530 M3=M3+M2
532 X2=(((X+X1)/2)*(B[1]-L))/27
534 X3=X3+X2
540 GOSUB 1000
550 C1=C2=X2=M2=0
560 L=B[1]
570 M1=M
572 X1=X
580 J=J+1
585 IF J>C0 THEN 1002
590 GOTO 190
600 REM            CUT SECTION
610 C2=B[2]-B[3]
620 D1=C2+(R/2*C)/(C-S)
630 D2=C2+(R/2*C)/(C+S)
640 H1=D1+S
650 H2=D2+S
660 X=0.5*(C2*(D1+D2)+(R/2*(H1+H2)))
665 M=0
670 IF J=1 THEN 700
680 X2=(((X+X1)/2)*(B[1]-L))/27
690 X3=X3+X2
692 M2=(((M+M1)/2)*(B[1]-L))/27
694 M3=M3+M2
700 GOSUB 1000
710 C1=C2=X2=M2=0
720 L=B[1]
730 X1=X
732 M1=M
740 J=J+1
```

```
745 IF J>C0 THEN 1002
750 GOTO 190
1000 PRINT B[1];TAB10,C2;TAB20,C1;TAB30,X;TAB40,M;TAB50,X2;TAB60,M2
1001 RETURN
1002 PRINT
1003 PRINT
1004 PRINT "TOTAL EMBANKMENT ="M3"CUBIC YARDS."
1005 PRINT "TOTAL EXCAVATION ="X3"CUBIC YARDS."
1006 DISP "ENTER SIDE SLOPE AND ENDING STA.";
1007 PRINT
1010 STOP
1030 END
```

The mission of the PACIFIC NORTHWEST FOREST AND RANGE EXPERIMENT STATION is to provide the knowledge, technology, and alternatives for present and future protection, management, and use of forest, range, and related environments.

Within this overall mission, the Station conducts and stimulates research to facilitate and to accelerate progress toward the following goals:

1. Providing safe and efficient technology for inventory, protection, and use of resources.
2. Development and evaluation of alternative methods and levels of resource management.
3. Achievement of optimum sustained resource productivity consistent with maintaining a high quality forest environment.

The area of research encompasses Oregon, Washington, Alaska, and, in some cases, California, Hawaii, the Western States, and the Nation. Results of the research will be made available promptly. Project headquarters are at:

Fairbanks, Alaska	Portland, Oregon
Juneau, Alaska	Olympia, Washington
Bend, Oregon	Seattle, Washington
Corwallis, Oregon	Wenatchee, Washington
La Grande, Oregon	

Mailing Address: ***Pacific Northwest Forest and Range Experiment Station***
P.O. Box 3141
Portland, Oregon 97208

GPO 991-518

The FOREST SERVICE of the U. S. Department of Agriculture is dedicated to the principle of multiple use management of the Nation's forest resources for sustained yields of wood, water, forage, wildlife, and recreation. Through forestry research, cooperation with the States and private forest owners, and management of the National Forests and National Grasslands, it strives — as directed by Congress -- to provide increasingly greater service to a growing Nation.